011764-502-M

30 July 1973

MEMO TO: File

FROM: Soon K. Cho

SUBJECT: RAM1D

1. Introduction

When an E- or H-polarized plane wave is incident normally upon a convex cylindrical body of general surface impedance, the electric current induced on the surface for each polarization can be described by an integral equation of the second kind. When the plane wave incidence is not normal but oblique, there then results a pair of coupled integral equations (Andreasen, 1965).

A computer program, RAM1D, has been written for an E- or H-polarized normal plane wave incidence, to compute the bistatic or monostatic scattering cross section of a convex cylindrical body via an approximate solution of the integral equation of the surface current. A brief discussion is given in Section 2 for the integral equations involved.

RAM1B deals with a pair of coupled integral equations for an oblique plane wave incidence, involving second derivatives of the Green's functions as kernels, even though the obliqueness in incidence is not carried through but abandoned in the end. It is worth noting that a second derivative of the Green's function involved in RAM1B gives rise to a singularity which is not integrable in the Riemann sense. A recognition of these factors in RAM1B motivated RAM1D.

For a special case where an E-polarized normal plane incident wave illuminates a perfectly conducting convex cylindrical body, the integral equation

DISTRIBUTION

Knott	Laxpati	Hiatt/File
Liepa	Cho	Senior

11764-502-M = RL-2240

of the surface current is reduced to that of the first kind with a Green's function as its kernel. As a consequence of the symmetry possessed by the Green's function with respect to the source and the observation coordinates, the kernel matrix derived from the integral equation through algebratization is symmetric. Although this fact is not exploited in RAM1D for this special case, a user could easily implement it with a slight modification to SUBROUTINE MATRIX, if so desired.

For a perfectly conducting body, the computation of the surface impedance of the body can be dispensed with altogether and this is done in RAM1D. The computation of the Hankel's functions involved is based on the polynomial approximation (Abramowitz, Segun, 1968).

In Section 3, a scheme used in computing the geometric factors of interest in solving the algebraic equation is presented in some detail. Although the computation of the set of end points of the cells of the body contour is not needed for our present case, the formula for the computation of the cell end points is included in the analysis for a future reference.

Finally, in Section 4, RAM1D listing is given, including the time tally of each SUBROUTINE for a sample calculation.

2. Integral Equations of Surface Current Density Functions

For a normal plane incident wave of the E- or H-polarization, the surface electric current induced on a convex cylindrical body of general surface impedance is given by an integral equation of the second kind (Senior, Knott, 1973)*. Thus, for s, $s' \in C$, the body contour (see Fig. 1), we have

H-polarization:
$$-H_z^i(s) = \frac{1}{2} K_s(s) + \frac{i}{4} \oint_C \left[\hat{r} \cdot \hat{n}(s') H_1^{(1)}(k|\bar{r}|) \right] K_s(s') \, d(ks') +$$
$$+ \frac{1}{4} \oint_C \left[\eta^{(1)}(s') \, H_0^{(1)}(k|\bar{r}|) \right] K_s(s') \, d(ks') \quad , \qquad (2.1)$$

*Note typographic sign errors in Eqs. (3-1) and (3-2) of this reference.

E-polarization:

$$Y_o E_z^i(s) = \frac{1}{2}\eta^{(2)}(s)\,K_z(s) + \frac{i}{4}\int_C \left[\eta^{(2)}(s)\hat{r}\cdot\hat{n}(s')\,H_1^{(1)}(k|\bar{r}|)\right] K_z(s')\,d(ks') +$$

$$+\frac{1}{4}\oint_C H_o^{(1)}(k|\bar{r}|)\,K(s')\,d(ks') \quad , \qquad (2.2)$$

where

$$|\bar{r}| = |\bar{\rho}(s) - \bar{\rho}(s')| \quad ,$$

$$\hat{r} = \frac{\bar{\rho}(s)-\bar{\rho}(s')}{|\bar{\rho}(s)-\bar{\rho}(s')|} \quad ,$$

$\hat{n}(s')$: the outward unit normal vector at $s = s'$,

$\int$: principal value integral.



Fig. 1: Geometry of a convex cylindrical body contour. $(\hat{n}, \hat{\tau}, \hat{z})$ is the local right-hand cartesian coordinate system.

Following the usual procedure, Eqs. (2.1) and (2.2) are algebratized in the forms

$$-H_s^i(s_i) = \sum_{j=1}^{M} A_{ij} K(s_j), \qquad i = 1, 2, 3, \ldots, M \tag{2.3}$$

where M denotes the number of the cells of C and s_i the midpoint of the ith cell, Δ_i, and

$$A_{ij} = \left[\frac{1}{2}\delta_{ij} + \frac{i}{4} r_{ij} \cdot n(s_j) H_1^{(1)}(k\, r_{ij})(1-\delta_{ij})\right] k\Delta_j + \\ + \eta^{(1)}(s_j)\left\{\left[1 + i\frac{2}{\pi}(\ell n \frac{k\Delta_j}{2} - 0.4228)\right]\delta_{ij} + \\ + H_o^{(1)}(k|\overline{r}_{ij}|)(1-\delta_{ij})\right\}\frac{k\Delta_j}{4}, \tag{2.4}$$

$$i, j = 1, 2, 3, \ldots, M;$$

and

$$Y_o E_z^i(s_i) = \sum_{j=1}^{M} B_{ij} K(s_j), \qquad i = 1, 2, 3, \ldots, M. \tag{2.5}$$

where

$$B_{ij} = \eta^{(2)}(s_j)\left[\frac{1}{2}\delta_{ij} + \frac{i}{4}\hat{r}_{ij} \cdot \hat{n}(s_j) H_1^{(1)}(k|\overline{r}_{ij}|)(1-\delta_{ij})\right] k\Delta_j + \\ + \left\{\left[1 + i\frac{2}{\pi}\ell n(\ell n\left(\frac{k\Delta_j}{2}\right) - 0.4228)\right]\delta_{ij} + \\ + H_o^{(1)}(k|\overline{r}_{ij}|)(1-\delta_{ij})\right\}\frac{k\Delta_j}{4}, \tag{2.6}$$

$$i, j = 1, 2, 3, \ldots, M;$$

$$\overline{r}_{ij} = \left|\overline{\rho}(s_i) - \overline{\rho}(s_j)\right| ,$$

$$\delta_{ij} = \begin{cases} 0 & \text{for } i \neq j, \\ 1 & \text{for } i = j, \quad i, j = 1, 2, 3, \ldots, M. \end{cases}$$

Equations (2.4) and (2.5) show that, as long as $\eta^{(2)}(s_j) \neq 0$, by replacing $\eta^{(2)}(s_j)$ by $1/\eta^{(1)}(s_j)$,

$$\eta^{(1)}(s_j) B_{ij} = A_{ij} \, , \tag{2.7}$$

demonstrating that, as pointed out by Knott and Senior (Knott, Senior, 1973), the use of the duality relations between $\overline{E}$, $\overline{H}$, μ and ϵ enables one to infer the solution of one integral equation from that of the other, for an imperfectly conducting body.

3. Geometry of the Contour of a Convex Cylindrical Body

Let the z-axis of the (x, y, z) Cartesian coordinate system be parallel to the axis of a convex cylinder and let C denote the contour of the cylinder in any fixed z-plane. Next, let C be decomposed into a set of segments of circular arcs of various radii of curvature. In the special limiting case where the segment is a straight line, we treat it separately. Thus, assuming that the contour of the convex cylinder, C, consists of a set of segments of circular arcs of finite radii of curvature and straight lines, in general, we wish to derive the formulas for the geometric quantities involved in solving the integral equations (2.1) and (2.2). If a segment is simulated by a circular arc, we assume that the Cartesian coordinates of the end points of the arc and the angle subtended by the arc are known; if the segment is a straight line we only need to specify the coordinates of the end points of the line.

We further assume that the contour C is symmetric with respect to the x-axis, say. This last assumption is made solely so as to enable us to specify the surface impedance along the contour in a convenient manner. Then it is sufficient to consider only the part of the contour in the upper half-plane.

A. Circular Arc Segments

Consider a circular arc segment, S. Let its end points $w_a = (x_z, y_a)$, $w_b = (x_b, y_b)$ and the angle subtended by S, θ, be given. Let $w_c = (x_c, y_c)$ denote the center of curvature of S. We divide S into N equal subarcs (or cells) and consider a jth subarc, Δs_j. Let (x_j, y_j) denote the midpoint of Δs_j and (X_j, Y_j) and (X_{j+1}, Y_{j+1}) its left end and the right end points. Let $\hat{n}_j$ denote the outward unit normal vector at (x_j, y_j).

Given w_a, w_b and θ, we wish to derive the formulas for a set of the unit outward normal vectors, the midpoints, the arc lengths and the end points of N subarcs. See Fig. 2.



Fig. 2: Geometry of a segment of circular arc.

Let d denote the chord length connecting w_a, w_b. Then,

$$d = \sqrt{(x_b - x_a)^2 + (y_b - y_a)^2}$$

$$= 2\, r \sin\frac{\theta}{2} \, , \qquad (3.1)$$

where r is the radius of curvature of S.

Hence

$$r = d/2 \sin\frac{\theta}{2} \, . \qquad (3.2)$$

Let $\Delta x = x_b - x_a$, $\Delta y = y_b - y_a$. Define τ by

$$\sin\tau = \Delta y/d \, , \quad \cos\tau = \Delta x/d.$$

Then, $w_c = (x_c, y_c)$ is given by

$$x_c = x_a + r\sin(\frac{\theta}{2} + \tau) \, ,$$

$$y_c = y_a - r\cos(\frac{\theta}{2} + \tau) \, ;$$

or

$$\left.\begin{aligned} x_c &= x_a + \frac{1}{2}(\Delta x + \Delta y \cot\frac{\theta}{2}) \, ; \\ y_c &= y_a + \frac{1}{2}(\Delta y - \Delta x \cot\frac{\theta}{2}) \, . \end{aligned}\right\} \quad \theta \neq 0. \qquad (3.3)$$

The jth subarc, Δs_j, is bounded by (X_j, Y_j) and (X_{j+1}, Y_{j+1}). Its midpoint is (x_j, y_j). Let q_j denote the angle subtended by the arc extending to the point (x_j, y_j) from w_a and let $\Delta\theta$ denote the angle subtended by Δs_j. Then, from Fig. 2, it is seen that

$$q_j = (2j - 1)\,\frac{\Delta\theta}{2} \, ,$$

$$Q_j = (j-1)\,\Delta\theta \, ; \qquad j = 1, 2, 3, \ldots, N \qquad (3.4)$$

where Q_j is the angle subtended by the arc extending to the point (X_j, Y_j) from w_a. We now have

$$x_j = x_c - r \sin\left(\frac{\theta}{2} + \tau - q_j\right) ,$$

$$y_j = y_c + r \cos\left(\frac{\theta}{2} + \tau - q_j\right), \quad j = 1, 2, 3, \ldots, N.$$

Eliminating x_c, y_c by use of Eq. (3.3) from the above relations, we obtain the coordinates of the midpoint of Δj:

$$\begin{aligned} x_j &= x_a + \frac{1}{2}\left[(\Delta x + \Delta y \cot\frac{\theta}{2})(1 - \cos q_j) - (\Delta y - \Delta x \cot\frac{\theta}{2}) \sin q_j\right] , \\ y_j &= y_a + \frac{1}{2}\left[(\Delta x + \Delta y \cot\frac{\theta}{2}) \sin q_j + (\Delta y - \Delta x \cot\frac{\theta}{2})(1 - \cos q_j)\right] , \\ &\theta \neq 0, \\ q_j &= (2j-1)\frac{\Delta\theta}{2}; \quad j = 1, 2, 3, \ldots, N. \end{aligned} \tag{3.5}$$

Replacing q_j in Eq. (3.5) by Q_j, we obtain the coordinates of the left end point of Δj:

$$\begin{aligned} X_j &= x_a + \frac{1}{2}\left[(\Delta x + \Delta y \cot\frac{\theta}{2})(1 - \cos Q_j) - (\Delta y - \Delta x \cot\frac{\theta}{2}) \sin Q_j\right] , \\ Y_j &= y_a + \frac{1}{2}\left[(\Delta x + \Delta y \cot\frac{\theta}{2}) \sin Q_j + (\Delta y - \Delta x \cot\frac{\theta}{2})(1 - \cos Q_j)\right] , \\ &\theta \neq 0, \\ Q_j &= (j-1)\Delta\theta ; \qquad j = 1, 2, 3, \ldots, N. \end{aligned} \tag{3.6}$$

We note that $X_{N+1} = x_b$, $Y_{N+1} = y_b$.

Let us now consider the outward unit normal vector at (x_j, y_j) of Δs_j. See Fig. 3.



Fig. 3: The outward unit normal vector at the midpoint of the jth subarc (or cell), ΔS_j.

From Fig. 3, we get the x, y components of $\hat{n}_j$ as

$$n_{jx} = -\sin\left(\frac{\theta}{2}+\tau-q_j\right) ,$$

$$n_{jy} = \cos\left(\frac{\theta}{2}+\tau-q_j\right) ,$$

or

$$n_{jx} = -\frac{\sin\frac{\theta}{2}}{d}\left[\left(\Delta x+\Delta y\cot\frac{\theta}{2}\right)\cos q_j + \left(\Delta y-\Delta x\cot\frac{\theta}{2}\right)\sin q_j\right] ,$$

$$n_{jy} = \frac{\sin\frac{\theta}{2}}{d}\left[\left(\Delta x+\Delta y\cot\frac{\theta}{2}\right)\sin q_j - \left(\Delta y-\Delta x\cot\frac{\theta}{2}\right)\cos q_j\right] , \tag{3.7}$$

$\theta \neq 0,$

$q_j = (2j-1)\frac{\Delta\theta}{2}; \qquad j = 1, 2, 3, \ldots, N.$

The length of the arc extending to (x_j, y_j) from w_a, which we denote by S_j, is

$$S_j = rq_j = r(2j-1)\frac{\Delta\theta}{2} , \qquad (3.8)$$

$$j = 1, 2, 3, \ldots, N,$$

and the length of each subarc (or cell) is $r\Delta\theta$.

B. Linear Segments

When the segment is a straight line, we assume that the end points w_a, w_b are given. Note that θ as defined in 3.A is zero in this case. Referring to Fig. 4, it is seen that

$$S = d = \sqrt{(\Delta x)^2 + (\Delta y)^2} , \qquad (3.9)$$

$$\Delta s_j = \frac{d}{N} , \quad j = 1, 2, 3, \ldots, N. \qquad (3.10)$$



Fig. 4: Geometry of a straight line segment.

Since

$$\sin\tau = \frac{\Delta y}{d} = \frac{y_j - y_a}{s_j} ,$$

$$\cos\tau = \frac{\Delta x}{d} = \frac{x_j - x_a}{s_j} ,$$

we get

$$x_j = x_a + \frac{s_j}{d}\Delta x \quad ,$$

$$y_j = y_a + \frac{s_j}{d}\Delta y \quad ,$$

$$s_j = (2j-1)\frac{d}{2N}; \qquad j = 1, 2, 3, \ldots, N. \tag{3.11}$$

Noting that $\frac{s_j}{d} = \frac{2j-1}{2N}$, we can write the coordinates of the midpoint of the jth cell, Δs_j, as

$$\left.\begin{aligned} x_j &= x_a + \frac{\Delta x}{2N}(2j-1) \quad , \\ y_j &= y_a + \frac{\Delta y}{2N}(2j-1) \quad , \end{aligned}\right\} \qquad j = 1, 2, 3, \ldots, N \tag{3.12}$$

$$\theta = 0.$$

For the outward unit normal vector at (x_j, y_j) of Δx_j, we have

$$\left.\begin{aligned} n_{jx} &= \cos\tau = \frac{\Delta x}{d} \ , \\ n_{jy} &= -\sin\tau = -\frac{\Delta y}{d} \ , \end{aligned}\right\} \qquad j = 1, 2, 3, \ldots, N. \tag{3.13}$$

The coordinates of the left end point of the jth cell are

$$\left.\begin{aligned} X_j &= x_j - \frac{\Delta x}{2N} \ , \\ Y_j &= y_j - \frac{\Delta y}{2N} \ , \end{aligned}\right\} \qquad j = 1, 2, 3, \ldots, N. \tag{3.14}$$

$$\theta = 0.$$

We note that

$$X_{N+1} = x_b, \quad Y_{N+1} = y_b \ .$$

4. List of RAM1D

The list of RAM1D, along with the time tally of the program, is given here. RAM1D took 23.8 seconds to compile in IBM 360 and for a sample case of a perfectly conducting ogive the contour of which was divided into 48 cells, the CPU time for evaluation of bistatic cross section at 37 points was 31.7 seconds for the case of H-polarization.

A. RAM1D

```
C..... MAIN PROGRAM...
C      THE ENTIRE ANALYSIS IS MADE IN THE CARTESIAN COORDINATE SYSTEM
C      NORMALIZED WITH RESPECT TO THE WAVELENGTH.
       COMMON/BOA/AK,AK2,AK4,RADIAN,RADINV,M,NSEG,IAT
       COMMON/BOB/X(50),Y(50),XN(50),YN(50),DSQ(50),S(50),ZS(50)
       COMMON/BOC/PINC(50),PHI(50),AA(50,51),AMAG(50),PHASE(50)
       COMMON/BOD/IPRINT,IOPT,IPP,IRSC,ALPHA
       COMPLEX PINC,PHI,AA,ZS
       IAT=1
       RADIAN=57.29578
       RADINV=0.017453
       AK4=1.570796
       AK2=3.1415927
       AK=6.2831846
       READ(5,100) M,NSEG,ALPHA
       READ(5,200) IOPT,IPP,IPRINT,IRSC
       IF(IOPT.EQ.0) GO TO 30
       IF(IPP.GE.2) GO TO 20
       DO 10 I=1,M
    10 READ(5,300)X(I),Y(I),XN(I),YN(I),DSQ(I),S(I),ZS(I)
       GO TO 30
    20 CONTINUE
       DO 15 I=1,M
    15 READ(5,350)X(I),Y(I),XN(I),YN(I),DSQ(I)
    30 CONTINUE
       CALL GEOM
       CALL MATRIX
       CALL SCATT
   100 FORMAT(2I5,F10.5)
   200 FORMAT(4I3)
   300 FORMAT(7F7.2)
   350 FORMAT(5F7.2)
       STOP
       END
```

```
      SUBROUTINE GEOM
C.....THIS SUBROUTINE COMPUTES BOTH GEOMETRIC FACTORS OF INTEREST
C     AND THE SURFACE IMPEDANCE WHEN THE BODY IS IMPERFECTLY CONDUCTING.
C     THE BODY CONTOUR,ASSUMED TO BE SYMMETRIC ABOUT THE X-AXIS,SAY,IS
C     CONSIDERED AS COMPOSED OF A SET OF CIRCULAR SEGMENTS.NOTE THAT
C     A STRAIGHT LINE IS A SPECIAL CASE OF A CIRCULAR ARC.
      COMMON/BOA/AK,AK2,AK4,RADIAN,RADINV,M,NSEG,IAT
      COMMON/BOB/X(50),Y(50),XN(50),YN(50),DSQ(50),S(50),ZS(50)
      COMMON/BOC/PINC(50),PHI(50),AA(50,51),AMAG(50),PHASE(50)
      COMMON/BOD/IPRINT,IOPT,IPP,IRSC,ALPHA
      DIMENSION LL(5)
      COMPLEX PINC,PHI,AA
      COMPLEX ZA,ZB,ZFAC,ZS
      ISTART=1
      DO 99 NU=1,NSEG
      LL(NU)=NU
C     NOTE THAT XA,YA,,XB,YB ARE NORMALIZED QUANTITIES
C     WITH RESPECT TO THE WAVELENGTH.
      READ(5,100) MM,XA,YA,XB,YB,ANG
      MMD=2*MM
      IEND=MM
      IF(NU.GT.1) GO TO 5
      GO TO 6
    5 ISTART=I+ISTART
      IEND=I+IEND
    6 IF(IPP.GE.2) GO TO 10
      READ(5,150) IZFORM,ZA,ZB,ZFAC,ZEX
      WRITE(6,400)
   10 DX=XB-XA
      DY=YB-YA
      CHORD=SQRT(DX*DX+DY*DY)
      IF(ANG.EQ.0.0) GO TO 20
      THETA2=0.5*ANG*RADINV
      ALFA=THETA2/MM
      DTHETA=2.0*ALFA
      SINB=SIN(THETA2)
      TANB=TAN(THETA2)
      IF(ANG.EQ.180.0)GO TO 25
      TREX=DX+DY/TANB
      TRAY=DY-DX/TANB
      GO TO 26
   25 TREX=DX
      TRAY=DY
   26 DARC=ALFA*CHORD/SINB
      DARC2=0.5*DARC
      GO TO 30
   20 DARC=CHORD/MM
      DARC2=0.5*DARC
   30 IF(IPP.GE.2) GO TO 33
      WRITE(6,450)
      GO TO 34
   33 WRITE(6,500)
   34 CONTINUE
      DO 75 I=ISTART,IEND
      FACT=2*I-1
```

```
      DSQ(I)=DARC
      IF(IPP.GE.2) GO TO 35
      S(I)=FACT*DARC2
      IF(IZFORM)40,45,50
   40 POWER=(1.-REAL(ZB))/ZEX
      POWER=0.5*POWER*POWER
      ZS(I)=ZA*EXP(-POWER)
      GO TO 35
   45 ZS(I)=ZFAC*(ZA+ZB*S(I)**ZEX)
      GO TO 35
   50 ZS(I)=ZFAC*(ZA+ZB*EXP(-S(I)*ZEX))
   35 IF(ANG.EQ.0.0)GO TO 60
      ARG1=FACT*ALFA
      SINQ=SIN(ARG1)
      COSQ=COS(ARG1)
      COSQ1=1.0-COSQ
      X(I)=XA+0.5*(TREX*COSQ1-TRAY*SINQ)
      Y(I)=YA+0.5*(TREX*SINQ+TRAY*COSQ1)
      XN(I)=-SINB*(TREX*COSQ+TRAY*SINQ)/CHORD
      YN(I)=SINB*(TREX*SINQ-TRAY*COSQ)/CHORD
      GO TO 70
   60 X(I)=XA+FACT*DX/MMD
      Y(I)=YA+FACT*DY/MMD
      XN(I)=-DY/CHORD
      YN(I)=DX/CHORD
   70 IF(IPP.GE.2)GO TO 74
      WRITE(6,200)LL(NU),I,X(I),Y(I),XN(I),YN(I),DSQ(I),S(I),ZS(I)
      GO TO 75
   74 WRITE(6,300)LL(NU),I,X(I),Y(I),XN(I),YN(I),DSQ(I)
   75 CONTINUE
      IF(IPP.GE.2)GO TO 80
      WRITE(6,460)
      GO TO 85
   80 WRITE(6,510)
   85 CONTINUE
      DO 99 I=ISTART,IEND
      IS=M-I+1
      IF(IPP.GE.2)GO TO 86
      S(IS)=S(I)
      ZS(IS)=ZS(I)
   86 DSQ(IS)=DSQ(I)
      X(IS)=X(I)
      Y(IS)=-Y(I)
      XN(IS)=XN(I)
      YN(IS)=-YN(I)
      IF(IPP.GE.2)GO TO 88
      WRITE(6,250)LL(NU),IS,X(IS),Y(IS),XN(IS),YN(IS),DSQ(IS),
     1            S(IS),ZS(IS)
      GO TO 99
   88 WRITE(6,350)LL(NU),IS,X(IS),Y(IS),XN(IS),YN(IS),DSQ(IS)
   99 CONTINUE
  100 FORMAT(I5,5F10.5)
  150 FORMAT(I3,7F7.2)
  200 FORMAT(5X,I5,2X,I3,8F7.3)
  250 FORMAT(6X,I5,2X,I3,2F7.3,1X,F7.3,1X,F7.3,2X,F7.3,1X,F7.3,2F7.3)
```

```
300 FORMAT(5X,I5,2X,I3,5F7.3)
350 FORMAT(6X,I5,2X,I3,2F7.3,1X,F7.3,1X,F7.3,2X,F7.3)
400 FORMAT(1H1,7HIZFORM=I3,2X,3HZA=2F7.2,2X,3HZB=2F7.2,
   1                            2X,5HZFAC=2F7.2,2X,4HZEX=F7.2/)
450 FORMAT(3X,7HSEGMENT,4X,1HI,3X,4HX(I),3X,4HY(I),2X,5HXN(I),
   1        2X,5HYN(I),1X,6HDSQ(I),3X,4HS(I),9X,5HZS(I)//)
500 FORMAT(3X,7HSEGMENT,4X,1HI,3X,4HX(I),3X,4HY(I),2X,5HXN(I),
   1      2X,5HYN(I),1X,6HDSQ(I)//)
460 FORMAT(1H1,3X,7HSEGMENT,3X,2HIS,2X,5HX(IS),2X,5HY(IS),2X,6HXN(IS),
   1      2X,6HYN(IS),2X,7HDSQ(IS),3X,5HS(IS),7X,7HZS(IS)//)
510 FORMAT(1H1,3X,7HSEGMENT,3X,2HIS,2X,5HX(IS),2X,5HY(IS),2X,6HXN(IS),
   1      2X,6HYN(IS),2X,7HDSQ(IS)//)
    RETURN
    END


    SUBROUTINE MATRIX
C.....THIS SUBROUTINE COMPUTES TWO MATRICES DERIVED FROM A SURFACE
C     CURRENT INTEGRAL EQUATION FOR A CONVEX CYLINDRICAL SCATTERER FOR
C     EITHER E-,OR,H-POLARIZATION:AN M BY 1 COLUMN MATRIX FOR THE
C     NORMAL PLANE WAVE INCIDENCE AND AN M BY M COEFFICIENT MATRIX
C     OF THE CURRENT MATRIX.
    COMMON/BOA/AK,AK2,AK4,RADIAN,RADINV,M,NSEG,IAT
    COMMON/BOB/X(50),Y(50),XN(50),YN(50),DSQ(50),S(50),ZS(50)
    COMMON/BOC/PINC(50),PHI(50),AA(50,51),AMAG(50),PHASE(50)
    COMMON/BOD/IPRINT,IOPT,IPP,IRSC,ALPHA
    COMPLEX PINC,PHI,AA
    COMPLEX ZS,BMA1,BMA2,CK4
    CK4=CMPLX(0.,AK4)
    ARG1=COS(ALPHA*RADINV)
    ARG2=SIN(ALPHA*RADINV)
    WRITE(6,500)
    DO 100 I=1,M
    ARG3=AK*(X(I)*ARG1+Y(I)*ARG2)
    PINC(I)=CMPLX(COS(ARG3),-SIN(ARG3))
    T1=REAL(PINC(I))
    T2=AIMAG(PINC(I))
    PH=ATAN2(T2,T1)*RADIAN
    WRITE(6,510)I,T1,T2,PH
    DO 100 J=1,M
    IF(I.EQ.J)GO TO 10
    DX=X(I)-X(J)
    DY=Y(I)-Y(J)
    RD=SQRT(DX*DX+DY*DY)
    CNR=(DX*XN(J)+DY*YN(J))/RD
    RANS=AK*RD
    CALL HANK(RANS,0,BJ0,BY0)
    BMA2=AK4*DSQ(J)*CMPLX(BJ0,BY0)
    CALL HANK(RANS,1,BJ1,BY1)
    BMA1=CK4*CNR*DSQ(J)*CMPLX(BJ1,BY1)
    GO TO 20
```

```
   10 BMA1=CMPLX(0.5,0.0)
      BMA2=DSQ(J)*CMPLX(AK4,ALOG(AK4*DSQ(J))-0.4228)
   20 IF(IPP.EQ.3)GO TO 25
      IF(IPP.EQ.0.OR.IPP.EQ.2)GO TO 30
      AA(I,J)=BMA1+BMA2*ZS(J)
      GO TO 100
   25 AA(I,J)=BMA1
      GO TO 100
   30 IF(IPP.EQ.2)GO TO 35
      AA(I,J)=BMA1*ZS(J)+BMA2
      GO TO 100
   35 AA(I,J)=BMA2
  100 CONTINUE
  500 FORMAT(1H1,10X,1HI,5X,20HPLANE WAVE INCIDENCE/)
  510 FORMAT(9X,I3,3X,3F7.2)
      RETURN
      END
```

```
      SUBROUTINE SCATT
C.....THIS SUBROUTINE COMPUTES BOTH THE SURFACE CURRENT DISTRIBUTION
C     AND EITHER A BISTATIC,OR BACK SCATTERING CROSS SECTION .
      COMMON/BOA/AK,AK2,AK4,RADIAN,RADINV,M,NSEG,IAT
      COMMON/BOB/X(50),Y(50),XN(50),YN(50),DSQ(50),S(50),ZS(50)
      COMMON/BOC/PINC(50),PHI(50),AA(50,51),AMAG(50),PHASE(50)
      COMMON/BOD/IPRINT,IOPT,IPP,IRSC,ALPHA
      DIMENSION PTS(100,6),A(6)
      DATA A/1H-,1H-,1H-,1HR,1HX,1H*/
      COMPLEX PINC,PHI,AA
      COMPLEX SUM,ZS,AMP
      IF(IRSC.EQ.1)GO TO 10
      READ(5,200) FIRST,FINAL,STEP
      NBIT=1+IFIX(ABS(FINAL-FIRST)/STEP)
      WRITE(6,210) IRSC,NBIT,FIRST,FINAL,STEP
      GO TO 20
```

```
10  WRITE(6,260)IRSC,ALPHA
20  CALL ZV08(AA,M,PINC,PHI,IAT)
    IF(IPP.EQ.0)WRITE(6,300)
    IF(IPP.EQ.1)WRITE(6,301)
    IF(IPP.EQ.2)WRITE(6,302)
    IF(IPP.EQ.3)WRITE(6,303)
30  WRITE(6,280)
    DO 35 I=1,M
    AMAG(I)=CABS(PHI(I))
    PHASE(I)=ATAN2(AIMAG(PHI(I)),REAL(PHI(I)))
35  WRITE(6,305)I,AMAG(I),PHASE(I)
    IF(IPPRINT.EQ.1)GO TO 40
    WRITE(6,12)
12  FORMAT(1H1)
    DO 13 I=1,6
    DO 13 J=1,M
    IF(I.EQ.1)PTS(J,I)=0.5
    IF(I.EQ.2)PTS(J,I)=1.0
    IF(I.EQ.3)PTS(J,I)=2.0
    IF(IPP.GE.2)GO TO 14
    IF(I.EQ.4)PTS(J,I)=0.5*REAL(ZS(J))+0.5
    IF(I.EQ.5)PTS(J,I)=0.5*AIMAG(ZS(J))+0.5
    GO TO 15
14  IF(I.EQ.4)PTS(J,I)=0.5
    IF(I.EQ.5)PTS(J,I)=0.5
15  CONTINUE
    IF(I.EQ.6)PTS(J,I)=AMAG(J)
13  CONTINUE
    CALL GPM(S,PTS,5,10,M,6,M,51,A)
40  IF(IRSC.EQ.1)GO TO 80
    WRITE(6,310)
    THETA =FIRST
    DO 70 I=1,NBIT
    RTHETA=THETA*RADINV
    ST=SIN(RTHETA)
    CT=COS(RTHETA)
    SUM=CMPLX(0.,0.)
    DO 60 J=1,M
    RDOTN=CT*XN(J)+ST*YN(J)
    ARG=-AK*(CT*X(J)+ST*Y(J))
    AMP=PHI(J)*DSQ(J)*CMPLX(COS(ARG),SIN(ARG))
    IF(IPP.EQ.3)GO TO 45
    IF(IPP.EQ.0.OR.IPP.EQ.2)GO TO 50
    SUM=SUM+AMP*(RDOTN-ZS(J))
    GO TO 60
45  SUM=SUM+AMP*RDOTN
    GO TO 60
50  IF(IPP.EQ.2)GO TO 55
    SUM=SUM+AMP*(RDOTN*ZS(J)-1.)
    GO TO 60
```

```
   55 SUM=SUM-AMP
   60 CONTINUE
      DBRSC=20.0*ALOG10(CABS(SUM))+1.987
      WRITE(6,350)I,THETA,DBRSC
   70 THETA=THETA+STEP
      GO TO 999
   80 WRITE(6,400)
      RALPHA=ALPHA*RADINV
      ST=SIN(RALPHA)
      CT=COS(RALPHA)
      SUM=CMPLX(0.,0.)
      DO 110 J=1,M
      RDOTN=CT*XN(J)+ST*YN(J)
      ARG=-AK*(CT*X(J)+ST*Y(J))
      AMP=PHI(J)*DSQ(J)*CMPLX(COS(ARG),SIN(ARG))
      IF(IPP.EQ.3)GO TO 85
      IF(IPP.EQ.0.OR.IPP.EQ.2)GO TO 90
      SUM=SUM+AMP*(RDOTN-ZS(J))
      GO TO 110
   85 SUM=SUM+AMP*RDOTN
      GO TO 110
   90 IF(IPP.EQ.2)GO TO 95
      SUM=SUM+AMP*(RDOTN*ZS(J)-1.0)
      GO TO 110
   95 SUM=SUM-AMP
  110 CONTINUE
      DBRSC=20.0*ALOG10(CABS(SUM))+1.987
      WRITE(6,450)ALPHA,DBRSC
  999 CONTINUE
  200 FORMAT(3F10.5)
  210 FORMAT(1H1,5X,5HIRSC=I3,2X,5HNBIT=I3,2X,6HFIRST=F10.5,
     1       2X,6HFINAL=F10.5,2X,5HSTEP=F10.5)
  260 FORMAT(1H1,5X,5HIRSC=I3,2X,6HALPHA=F10.5)
  300 FORMAT(2X,51HCURRENT DISTRIBUTION FOR E-LOLARIZATION,NON-ZERO ZS/)
  301 FORMAT(2X,51HCURRENT DISTRIBUTION FOR H-LOLARIZATION,NON-ZERO ZS/)
  302 FORMAT(5X,44HCURRENT DISTRIBUTION FOR E-POLARIZATION,ZS=0/)
  303 FORMAT(5X,44HCURRENT DISTRIBUTION FOR H-POLARIZATION,ZS=0/)
  280 FORMAT(2X,11HCELL NUMBER,2X,7HAMAG(I),2X,8HPHASE(I)/)
  305 FORMAT(7X,I3,5X,F7.2,3X,F7.2)
  310 FORMAT(1H1,4X,11HREC. POINTS,5X,5HTHETA,3X,23HBISTATIC CROSS SECTI
     1ON/)
  350 FORMAT(4X,I3,11X,F7.2,10X,E12.4)
  400 FORMAT(1H1,5X,10HINC. ANGLE,2X,28HBACKSCATTERING CROSS SECTION/)
  450 FORMAT(6X,F10.5,15X,E12.4)
      RETURN
      END
```

```
      SUBROUTINE ZVO8 (A,N,X,Y,IAT)
      DIMENSION A(50,51),X(50),Y(50),L(50),M(50)
      COMPLEX A,D,BIGA,HOLD,Y,X
      INTEGER L,M
      IF(IAT-1) 200,200,300
  200 CONTINUE
      D=CMPLX(1.0,0.0)
      DO 80 K=1,N
      L(K)=K
      M(K)=K
      BIGA=A(K,K)
      DO 20 J=K,N
      DO 20 I=K,N
   10 IF (C ABS(BIGA)-C ABS(A(I,J))) 15,20,20
   15 BIGA=A(I,J)
      L(K)=I
      M(K)=J
   20 CONTINUE
      J= L(K)
      IF (J-K) 35,35,25
   25 DO 30 I=1,N
      HOLD=-A(K,I)
      A(K,I)=A(J,I)
   30 A(J,I)=HOLD
   35 I=M(K)
      IF (I-K)  45,45,38
   38 DO 40  J=1,N
      HOLD=-A(J,K)
      A(J,K)=A(J,I)
   40 A(J,I)=HOLD
   45 IF (C ABS(BIGA))  48,46,48
   46 D=CMPLX(0.0,0.0)
      RETURN
   48 DO 55 I=1,N
      IF (I-K) 50,55,50
   50 A(I,K)=A(I,K)/(-BIGA)
   55 CONTINUE
C  REDUCE MATRIX
      DO 65 I=1,N
      DO 65 J=1,N
      IF (I-K) 60,65,60
   60 IF (J-K) 62,65,62
   62 A(I,J)=A(I,K)*A(K,J)  +A(I,J)
   65 CONTINUE
C  DIVIDE ROW BY PIVOT
      DO 75 J=1,N
      IF (J-K) 70,75,70
   70 A(K,J)=A(K,J)/BIGA
   75 CONTINUE
C  PRODUCT OF PIVOTS
      D=D*BIGA
      A(K,K)=(1.000,0.000) /BIGA
   80 CONTINUE
      BN=N
      DMAG=C ABS(D)*(2.**BN)
```

```
      K=N
100   K=K-1
      IF (K)  150,150,105
105   I=L(K)
      IF (I-K)  120,120,108
108   DO 110 J=1,N
      HOLD=A(J,K)
      A(J,K)=-A(J,I)
110   A(J,I)=HOLD
120   J=M(K)
      IF (J-K)  100,100,125
125   DO 130 I=1,N
      HOLD=A(K,I)
      A(K,I)=-A(J,I)
130   A(J,I)=HOLD
      GO TO 100
150   CONTINUE
300   CONTINUE
      DO 210 I=1,N
      Y(I)=CMPLX(0.0,0.0)
      DO 210 J=1,N
210   Y(I)=A(I,J)*X(J)+Y(I)
      RETURN
      END
```

```
      SUBROUTINE HANK(R,N,BJ,BY)
      MP=0
      DIMENSION A(7),B(7),C(7),D(7),E(7),F(7),G(7),H(7)
      DATA A,B,C,D,E,F,G,H/1.0,
     &-2.24999970, 1.26562080,-0.31638660,
     & 0.04444790,-0.00394440, 0.00021000,
     & 0.36746691, 0.60559366,-0.74350384,
     & 0.25300117,-0.04261214, 0.00427961,-0.00024845,
     & 0.50000000,-0.56249985, 0.21093573,
     &-0.03954289, 0.00443319,-0.00031761,-0.00001109,
     &-0.63661980, 0.22120910, 2.16827090,
     &-1.31648270, 0.31239510,-0.04009760, 0.00278730,
     & 0.79788456,-0.00000077,-0.00552740,
     &-0.00009512, 0.00137237,-0.00072805, 0.00014476,
     &-0.78539816,-0.04166397,-0.00003945,
     & 0.00262573,-0.00054125,-0.00029333, 0.00013558,
     & 0.79788456, 0.00000156, 0.01659667,
     & 0.00017105,-0.00249511, 0.00113653,-0.00020033,
     &-2.35619449, 0.12499612, 0.00005650,
     &-0.00637879, 0.00074348, 0.00079824,-0.00029165/
      IF(R.LE.0.0)GO TO 50
      IF(R.GT.3.0)GO TO 20
      X=R/3.0
      X=X*X
      IF(N.NE.0)GO TO 10
      CALL ADAM(A,X,Y)
      BJ=Y
      CALL ADAM(B,X,Y)
      BY=0.6366198*ALOG(0.5*R)*BJ+Y
      RETURN
   10 IF(N.NE.1)GO TO 60
      CALL ADAM(C,X,Y)
      BJ=R*Y
      CALL ADAM(D,X,Y)
      BY=0.6366198*ALOG(0.5*R)*BJ+Y/R
      RETURN
   20 X=3.0/R
      IF(N.NE.0)GO TO 30
      CALL ADAM(E,X,Y)
      GOOD=Y/SQRT(R)
      CALL ADAM(F,X,Y)
      GO TO 40
   30 IF(N.NE.1)GO TO 60
      CALL ADAM(G,X,Y)
      GOOD=Y/SQRT(R)
      CALL ADAM(H,X,Y)
   40 T=R+Y
      BJ=GOOD*COS(T)
      BY=GOOD*SIN(T)
      RETURN
   50 MP=MP+1
   60 MP=MP+1
      RETURN
      END
```

```
      SUBROUTINE ADAM(C,X,Y)
      DIMENSION C(7)
      Y=X*C(7)
      DO 10 I=1,5
   10 Y=X*(C(7-I)+Y)
      Y=Y+C(1)
      RETURN
      END
```

```
      SUBROUTINE GPM(X, Y, L, S, M, N, W, LN, A)
C  THIS SUBROUTENE HAS BEEN ESPECIALLY MODIFIEF FOR USE WITH RAM1B
C**
C**   CONTROL
C**
C
C**   CALL GPM(X, Y, L, S, M, N, W, LN, A)
C**
C**    WHERE
C**          X = ARRAY OF INDEPENDENT VALURS, DIMENSIONED X(M).
C**          Y = ARRAY OF SETS OF DEPENDENT VALUES, DIMENSIONED Y(M,N).
C**          L = NUMBER OF LINES TO BE SKIPPED BEFORE DISPLAY.
C**          S = NUMBER OF SPACES FROM LEFT SIDE OF PAGE TO
C**              BE SKIPPED BEFORE DISPLAY.
C**          M = NUMBER POINTS IN EACH SET.
C**          N = NUMBER OF SETS OF POINTS.
C**          W = WIDTH OF DISPLAY IN PRINT SPACES.
C**         LN = LENGTH OF DISPLAY IN PRINT LINES.
C**          A = ARRAY OF SINGLE CHARACTERS, DIMENSIONED A(N),TO
C**              REPRESENT THE TREND FOR EACH SET (EX.- DATA A/1HA,
C**              1HB,...ETC.)
C**
      DIMENSION X(M), A(N)
      DIMENSION PLOT(51,100),Y(100,6)
C
      INTEGER S, W, W1
C
      DATA BLANK/1H /,VERT/1HI/,HORIZ/1H=/
C
C**   CHECK MAXIMUM WIDTH AND LENGTH REQUESTED AND
C**   EXIT IF NOT CORRECT
C**
      IF (S+W .GT. 131) GO TO 900
      IF (L+LN .GT. 58) GO TO 800
C
C**   FIND MINIMUM AND MAXIMUM OF X AND Y
C**
      XMAX=X(1)
      XMIN=X(1)
C
      DO 10 I=2,M
      IF (X(I) .GT. XMAX) XMAX=X(I)
   10 IF (X(I) .LT. XMIN) XMIN=X(I)
C
      YMIN=0.
      YMAX=2.5
C
C**   COMPUTE SCALE FACTOR -- P FOR X, Q FOR Y
C**
      P=FLOAT(W-1)/(XMAX-XMIN)
      Q=FLOAT(LN-1)/(YMAX-YMIN)
C
C**   BLANK PLOT ARRAY
C**
      DO 30 I=1,W
```

```
      DO 30 J=1,LN
   30 PLOT(J,I)=BLANK
C
C**   CONSTRUCT BORDER OF DISPLAY
C**
      DO 40 J=1,LN
      I=1
      PLOT(J,I)=VERT
      I=W
   40 PLOT(J,I)=VERT
C
      W1=W-1
      DO 50 I=2,W1
      J=1
      PLOT(J,I)=HORIZ
      J=LN
   50 PLOT(J,I)=HORIZ
C
C**   COMPUTE SUBSCRIPTS AND INSERT TREND CHARACTER IN
C**   PLOT ARRAY
C**
      DO 60 I=1,M
      DO 60 J=1,N
      I1=1+INT(0.5+P*(X(I)-XMIN))
      J1=LN-INT(0.5+Q*(Y(I,J)-YMIN))
   60 PLOT(J1,I1)=A(J)
C
C**   SKIP L LINES BEFORE BEGINNING DISPLAY PRINTING
C
      DO 70 K=1,L
   70 PRINT 600
  600 FORMAT (1H )
C
C**   WRITE OUT PLOT ARRAY, SKIPPING S SPACES BEFORE PRINTING
C**   EACH LINE OF DISPLAY
C**
      DO 80 J=1,LN
   80 PRINT 601, (BLANK, K=1,S), (PLOT(J,I), I=1,W)
  601 FORMAT (132A1)
      PRINT 602, XMIN, XMAX, YMIN, YMAX
  602 FORMAT (1H0,5X,6HXMIN =E16.8,10X,6HXMAX =E16.8,10X,
     X      6HYMIN =E16.8,10X,6HYMAX =E16.8)
      RETURN
C**
C**   ERROR MESSAGES BEFORE TERMINATION
C**
  800 PRINT 603, L, LN
  603 FORMAT (30HAL+LN IS GREATER THAN 58    L =I3,5X,4HLN =I3)
      CALL SYSTEM
  900 PRINT 604, S, W
  604 FORMAT (30HAS+W IS GREATER THAN 131    S =I3,5X,3HW =I3)
      CALL SYSTEM
      RETURN
      END
```

B. Time Tally

```
-----------------------------+
SPRINT#  00000-00000   1.38  |
ALOG     00000-00000   0.34  |
CABS     00000-00000  23.10  |
CDVD#    00000-00000   1.55  |
CMPY#    00000-00000  12.76  |
COS      00000-00000   2.24  |
SIN      00000-00000   1.90  |
SQRT     00000-00000   1.55  |
-----------------------------+
#TIMTALY 00000-01BFF   0.00  |
#FAKEMTS 00000-00EFF   0.00  |
         00F00-00FFF   0.52  |
         01000-010FF   0.86  |
         01100-012FF   0.00  |
         01300-013FF   0.17  |
         01400-014FF  18.62  |
         01500-052FF   0.00  |
-----------------------------+
MAIN     00000-003FF   0.00  |
ADAM     00000-000FF   1.55  |
         00100-001FF   7.07  |
GEOM     00000-00CFF   0.00  |
         00D00-00DFF   0.17  |
         00E00-00EFF   0.00  |
MATRIX   00000-003FF   0.00  |
         00400-004FF   0.34  |
         00500-005FF   0.52  |
         00600-006FF   0.34  |
         00700-007FF   0.00  |
HANK     00000-000FF   0.52  |
         00100-001FF   0.00  |
         00200-002FF   1.72  |
         00300-003FF   2.41  |
         00400-004FF   1.55  |
         00500-005FF   0.17  |
SCATT    00000-019FF   0.00  |
ZV08     00000-004FF   0.00  |
         00500-005FF   2.93  |
         00600-006FF   0.86  |
         00700-007FF   3.45  |
         00800-008FF   6.90  |
         00900-009FF   0.34  |
         00A00-00AFF   0.52  |
         00B00-00BFF   0.52  |
GPM      00000-057FF   0.00  |
IBCOM#   00000-002FF   0.00  |
         00300-003FF   0.34  |
         00400-004FF   0.34  |
         00500-005FF   0.17  |
         00600-016FF   0.00  |
ADCON#   00000-000FF   0.00  |
FCVZO    00100-001FF   0.00  |
FCVLO    00200-004FF   0.00  |
FCVIO    00500-005FF   0.00  |
         00600-006FF   0.17  |
```

	00700-007FF	1.03
	00800-008FF	0.00
	00900-009FF	0.69
FCVEO	00A00-00BFF	0.00
FCVCO	00C00-00CFF	0.00
FCVTHB	00D00-010FF	0.00
FIOCS#	00000-001FF	0.00
	00200-002FF	0.17
	00300-004FF	0.00
	00500-005FF	0.17
	00600-00EFF	0.00